This Maple worksheet is denoted sirs3pra.
It is used to study an SIRS model with demography, with R0>1 and α1 large.
The deterministic version of the model shows damped oscillations toward an endemic infection level.
The particular model dealt with here has a state space with 3 variables: S, I, and R, and the infection rate is "modified proper", which means that the denominator of the infection rate equals S+I+R.
Two things are done here.
First we derive an expression for the angular frequency of the deterministic model oscillations, and after that we give a derivation of the moments of a diffusion approximation.
Ingemar Nåsell, KTH, Stockholm, 2012-10-03.

```
> restart;
  with(LinearAlgebra,Transpose,Eigenvalues,
  CharacteristicPolynomial);
  with(VectorCalculus,Jacobian);
  interface(imaginaryunit=II);
```

$$[Transpose, Eigenvalues, CharacteristicPolynomial]$$

$$[Jacobian]$$

$$I \qquad (1)$$

The reason for changing the notation used for the imaginary unit is that "I" will be used below to denote the number of infected individuals.
The original transition rates are stored in the table transA:

```
> transA:=table([[1,0,0]=mu*N,[1,0,-1]=delta*R,[-1,1,0]=beta*S*I/
  (S+I+R),[-1,0,0]=mu*S,[0,-1,1]=gamma*I,[0,-1,0]=mu*I,[0,0,-1]=
  mu*R]);
```

$$transA := table\left(\left[[1, 0, 0] = \mu N, [-1, 1, 0] = \frac{\beta S I}{S+I+R}, [1, 0, -1] = \delta R, [0, -1, 1] = \gamma I, [0, -1, 0] = \mu I, [-1, 0, 0] = \mu S, [0, 0, -1] = \mu R\right]\right) \qquad (2)$$

After scaling: x1=S/N, x2=I/N, x3=R/N, and reparametrization R0=β/(γ+μ), α1= (γ+μ)/μ,
α2=(δ+μ)/μ, we get: S=x1*N, I=x2*N, R=x3*N, β=μ*α1*R0, γ=μ*(α1−1), δ=μ*(α2−1).

The Maple procedure "scale" is used to rewrite the transition rates after this rescaling and reparametrization.

```
> scale:=proc(tab)
    local xA,n,xB,xC;
    xA:=op(2,eval(tab));
    n:=nops(xA);
    xB:=subs(S=x1*N,I=x2*N,R=x3*N,beta=mu*alpha1*R0,gamma=mu*
  (alpha1-1),delta=mu*(alpha2-1),xA);
    xC:=[seq(lhs(op(i,xB))=simplify(rhs(op(i,xB)/N)),i=1..n)];
    table(xC);
  end proc:
```

Apply the rescaling and reparametrization described above to get the new table of transition rates "trans":

```
> trans:=scale(transA);
```

$$trans := table\left(\left[[1, 0, 0] = \mu, [-1, 1, 0] = \frac{\mu\,\alpha 1\, R0\, x1\, x2}{x1 + x2 + x3}, [1, 0, -1] = \mu\,(\alpha 2 - 1)\, x3, [0, -1, 1] = \mu\,(\alpha 1 - 1)\, x2, [0, -1, 0] = \mu\, x2, [-1, 0, 0] = \mu\, x1, [0, 0, -1] = \mu\, x3\right]\right) \quad (3)$$

Next is a procedure that determines the right hand sides of the deterministic ODEs for the scaled variables x1, x2, x3 from the table of transition rates:

```
> equ:=proc(i,tab)
    local x,n;
    x:=op(2,eval(tab));
    add(lhs(x[n])[i]*rhs(x[n]),n=1..nops(x));
  end proc:
```

The 3 right-hand sides are as follows:

```
> eq1:=equ(1,trans);
  eq2:=simplify(equ(2,trans));
  eq3:=equ(3,trans);
```

$$eq1 := \mu - \frac{\mu\,\alpha 1\, R0\, x1\, x2}{x1 + x2 + x3} + \mu\,(\alpha 2 - 1)\, x3 - \mu\, x1$$

$$eq2 := \frac{\mu\, x2\, \alpha 1\,(R0\, x1 - x1 - x2 - x3)}{x1 + x2 + x3}$$

$$eq3 := -\mu\,(\alpha 2 - 1)\, x3 + \mu\,(\alpha 1 - 1)\, x2 - \mu\, x3 \quad (4)$$

Critical points:

```
> crit:=solve({eq1,eq2,eq3},{x1,x2,x3});
```

$$crit := \{x1 = 1, x2 = 0, x3 = 0\}, \left\{x1 = \frac{1}{R0}, x2 = \frac{\alpha 2\,(R0 - 1)}{R0\,(\alpha 2 + \alpha 1 - 1)}, x3 = \frac{(\alpha 1 - 1)\,(R0 - 1)}{R0\,(\alpha 2 + \alpha 1 - 1)}\right\} \quad (5)$$

The point corresponding to an endemic infection level is termed (x10,x20,x30):

```
> x10:=rhs(crit[2][1]);
  x20:=rhs(crit[2][2]);
  x30:=map(factor,rhs(crit[2][3]));
```

$$x10 := \frac{1}{R0}$$

$$x20 := \frac{\alpha 2\,(R0 - 1)}{R0\,(\alpha 2 + \alpha 1 - 1)}$$

$$x30 := \frac{(\alpha 1 - 1)\,(R0 - 1)}{R0\,(\alpha 2 + \alpha 1 - 1)} \quad (6)$$

The Jacobian of the system of ODEs is denoted Bx:

```
> Bx:=Jacobian([eq1,eq2,eq3],[x1,x2,x3]);
```

$$Bx := \left[\left[-\frac{\mu\,\alpha 1\,R0\,x2}{x1+x2+x3}+\frac{\mu\,\alpha 1\,R0\,x1\,x2}{(x1+x2+x3)^2}-\mu,\ -\frac{\mu\,\alpha 1\,R0\,x1}{x1+x2+x3}+\frac{\mu\,\alpha 1\,R0\,x1\,x2}{(x1+x2+x3)^2},\ \frac{\mu\,\alpha 1\,R0\,x1\,x2}{(x1+x2+x3)^2}+\mu\,(\alpha 2-1)\right],\right.$$
$$\left[\frac{\mu\,x2\,\alpha 1\,(R0-1)}{x1+x2+x3}-\frac{\mu\,x2\,\alpha 1\,(R0\,x1-x1-x2-x3)}{(x1+x2+x3)^2},\ \frac{\mu\,\alpha 1\,(R0\,x1-x1-x2-x3)}{x1+x2+x3}-\frac{\mu\,x2\,\alpha 1}{x1+x2+x3}-\frac{\mu\,x2\,\alpha 1\,(R0\,x1-x1-x2-x3)}{(x1+x2+x3)^2},\ -\frac{\mu\,x2\,\alpha 1}{x1+x2+x3}-\frac{\mu\,x2\,\alpha 1\,(R0\,x1-x1-x2-x3)}{(x1+x2+x3)^2}\right],$$
$$\left.\left[0,\ \mu\,(\alpha 1-1),\ -\mu\,(\alpha 2-1)-\mu\right]\right] \qquad (7)$$

Evaluate the Jacobian at the critical point:

```
> B:=simplify(subs(x1=x10,x2=x20,x3=x30,Bx));
```

$$B := \left[\left[-\frac{(\alpha 2\,\alpha 1\,R0^2-2\,\alpha 2\,\alpha 1\,R0+\alpha 2\,\alpha 1+R0\,\alpha 2+\alpha 1\,R0-R0)\,\mu}{R0\,(\alpha 2+\alpha 1-1)},\ -\frac{(\alpha 1\,R0-R0+\alpha 2)\,\mu\,\alpha 1}{R0\,(\alpha 2+\alpha 1-1)},\ \frac{(2\,\alpha 2\,\alpha 1\,R0-\alpha 2\,\alpha 1+R0\,\alpha 2^2-2\,R0\,\alpha 2-\alpha 1\,R0+R0)\,\mu}{R0\,(\alpha 2+\alpha 1-1)}\right],\right.$$
$$\left[\frac{\mu\,\alpha 2\,(R0-1)^2\,\alpha 1}{R0\,(\alpha 2+\alpha 1-1)},\ -\frac{\mu\,\alpha 2\,(R0-1)\,\alpha 1}{R0\,(\alpha 2+\alpha 1-1)},\ -\frac{\mu\,\alpha 2\,(R0-1)\,\alpha 1}{R0\,(\alpha 2+\alpha 1-1)}\right],$$
$$\left.\left[0,\ \mu\,(\alpha 1-1),\ -\mu\,\alpha 2\right]\right] \qquad (8)$$

We proceed to determine the eigenvalues of the matrix B.
We use first the command "Eigenvalues" to show that one of the eigenvalues equals -μ.
After that, we use the command "CharacteristicPolynomial" and the knowledge that one eigenvalue equals −μ to derive a quadratic equation for the remaining two eigenvalues.

```
> eig:=Eigenvalues(B);
```

$$eig := \left[\left[-\mu\right],\right. \qquad (9)$$

$$\left[\frac{1}{2}\,\frac{1}{\alpha 2+\alpha 1-1}\Big(\Big(-\alpha 2^2-\alpha 2\,\alpha 1\,R0+\alpha 2 + \big(\alpha 2^4-2\,\alpha 2^3\,\alpha 1\,R0-2\,\alpha 2^3+\alpha 2^2\,\alpha 1^2\,R0^2+6\,\alpha 2^2\,\alpha 1\,R0+\alpha 2^2 + 8\,\alpha 2^2\,\alpha 1^2-8\,\alpha 2^2\,\alpha 1+4\,\alpha 2^3\,\alpha 1-8\,\alpha 2^2\,\alpha 1^2\,R0+4\,\alpha 2\,\alpha 1^3-8\,\alpha 2\,\alpha 1^2 - 4\,\alpha 2\,\alpha 1^3\,R0+8\,\alpha 2\,\alpha 1^2\,R0+4\,\alpha 2\,\alpha 1-4\,\alpha 2\,\alpha 1\,R0\big)^{1/2}\Big)\,\mu\Big)\right],$$

$$\left[-\frac{1}{2}\,\frac{1}{\alpha 2+\alpha 1-1}\Big(\Big(\alpha 2^2+\alpha 2\,\alpha 1\,R0-\alpha 2 + \big(\alpha 2^4-2\,\alpha 2^3\,\alpha 1\,R0-2\,\alpha 2^3+\alpha 2^2\,\alpha 1^2\,R0^2+6\,\alpha 2^2\,\alpha 1\,R0+\alpha 2^2 + 8\,\alpha 2^2\,\alpha 1^2-8\,\alpha 2^2\,\alpha 1+4\,\alpha 2^3\,\alpha 1-8\,\alpha 2^2\,\alpha 1^2\,R0+4\,\alpha 2\,\alpha 1^3-8\,\alpha 2\,\alpha 1^2 - 4\,\alpha 2\,\alpha 1^3\,R0+8\,\alpha 2\,\alpha 1^2\,R0+4\,\alpha 2\,\alpha 1-4\,\alpha 2\,\alpha 1\,R0\big)^{1/2}\Big)\,\mu\Big)\right]\Bigg]$$

One of the eigenvalues is thus seen to be equal to $-\mu$

Next we determine the characteristic polynomial of the matrix B.

```
> p:=CharacteristicPolynomial(B,lambda);
```

$$p := \lambda^3+\frac{\mu\,(\alpha 2^2+\alpha 2\,\alpha 1\,R0+\alpha 1-1)\,\lambda^2}{\alpha 2+\alpha 1-1} + \frac{\alpha 2\,\mu^2\,(-\alpha 1^2+\alpha 1-\alpha 2\,\alpha 1+\alpha 2\,\alpha 1\,R0+\alpha 2-1+\alpha 1^2\,R0)\,\lambda}{\alpha 2+\alpha 1-1}+\mu^3\,\alpha 2\,(R0-1)\,\alpha 1 \qquad \textbf{(10)}$$

To proceed, we derive a quadratic equation for the remaining two eigenvalues by dividing p by $\lambda+\mu$, and simplifying:

```
> p1:=map(simplify,collect(simplify(p/(lambda+mu)),lambda));
```

$$p1 := \lambda^2+\frac{\mu\,\alpha 2\,(\alpha 2+\alpha 1\,R0-1)\,\lambda}{\alpha 2+\alpha 1-1}+(R0-1)\,\alpha 2\,\mu^2\,\alpha 1 \qquad \textbf{(11)}$$

Define R1 = $\frac{\alpha 1\cdot R0+\alpha 2-1}{\alpha 1+\alpha 2-1}$. The two roots of the equation p1=0 can then be written $-\frac{\mu\cdot\alpha 2\cdot R1}{2}\pm i\cdot\Omega$, where $\Omega=\mu\cdot\sqrt{\alpha 1\cdot\alpha 2\cdot(R0-1)-\left(\frac{\alpha 2\cdot R1}{2}\right)^2}$.

This finishes the study of the eigenvalues.

We proceed to determine approximations of the covariances of a diffusion approximation.

Covariances of x[i]x[j] are determined by cov1:

```
> cov1:=proc(i,j,tab)
    local x,n;
    x:=op(2,eval(tab));
    add(lhs(x[n])[i]*lhs(x[n])[j]*rhs(x[n]),n=1..nops(x));
  end proc:
```

The local covariance matrix S is determined by the procedure cov:

```
> cov:=proc(tab)
    local i,j,d,S;
    d:=nops(lhs(op(2,eval(tab))[1]));
    for i from 1 to d do
      for j from 1 to d do
        S[i,j]:=cov1(i,j,tab);
      od;
    od;
    S:=Matrix(d,S);
  end proc:
```

By using the table of transition rates in "trans", we get

```
> Sx:=simplify(cov(trans));
```

$$Sx := \left[\left[\frac{1}{x1+x2+x3}\left(\mu\left(x1+x2+x3+\alpha 1\, R0\, x1\, x2+x3\,\alpha 2\, x1+x3\,\alpha 2\, x2+x3^2\,\alpha 2 - x3\, x2 - x3^2 + x1^2 + x1\, x2\right)\right), -\frac{\mu\,\alpha 1\, R0\, x1\, x2}{x1+x2+x3}, -\mu\left(\alpha 2-1\right) x3\right], \left[-\frac{\mu\,\alpha 1\, R0\, x1\, x2}{x1+x2+x3}, \frac{\mu\, x2\,\alpha 1\left(R0\, x1+x1+x2+x3\right)}{x1+x2+x3}, -\mu\left(\alpha 1-1\right) x2\right], \left[-\mu\left(\alpha 2-1\right) x3, -\mu\left(\alpha 1-1\right) x2, \mu\, x3\,\alpha 2+\mu\, x2\,\alpha 1-\mu\, x2\right]\right] \quad (12)$$

Evaluate the local covariance matrix at the critical point:

```
> S:=simplify(subs(x1=x10,x2=x20,x3=x30,Sx));
```

$$S := \left[\left[\frac{2\left(-\alpha 2\,\alpha 1+\alpha 2\,\alpha 1\, R0+\alpha 2+\alpha 1-1\right)\mu}{R0\left(\alpha 2+\alpha 1-1\right)}, -\frac{\mu\,\alpha 2\left(R0-1\right)\alpha 1}{R0\left(\alpha 2+\alpha 1-1\right)}, -\frac{\mu\left(\alpha 2-1\right)\left(\alpha 1-1\right)\left(R0-1\right)}{R0\left(\alpha 2+\alpha 1-1\right)}\right], \left[-\frac{\mu\,\alpha 2\left(R0-1\right)\alpha 1}{R0\left(\alpha 2+\alpha 1-1\right)}, \frac{2\,\mu\,\alpha 2\left(R0-1\right)\alpha 1}{R0\left(\alpha 2+\alpha 1-1\right)}, -\frac{\mu\left(\alpha 1-1\right)\alpha 2\left(R0-1\right)}{R0\left(\alpha 2+\alpha 1-1\right)}\right], \left[-\frac{\mu\left(\alpha 2-1\right)\left(\alpha 1-1\right)\left(R0-1\right)}{R0\left(\alpha 2+\alpha 1-1\right)}, -\frac{\mu\left(\alpha 1-1\right)\alpha 2\left(R0-1\right)}{R0\left(\alpha 2+\alpha 1-1\right)}, \frac{2\,\mu\left(\alpha 1-1\right)\alpha 2\left(R0-1\right)}{R0\left(\alpha 2+\alpha 1-1\right)}\right]\right] \quad (13)$$

Now proceed to solve A=-S, where A=B*SIG+SIG*BT, and where BT=Transpose(B).
First introduce notation for the elements of the matrix SIG:

```
> SIG:=Matrix(3,[s11,s12,s13,s21,s22,s23,s31,s32,s33]);
```

$$SIG := \begin{bmatrix} s11 & s12 & s13 \\ s21 & s22 & s23 \\ s31 & s32 & s33 \end{bmatrix} \quad (14)$$

Next, evaluate the matrix A:

```
> A:=Matrix(evalm(B&*SIG+SIG&*Transpose(B))):
```

Solve the 9 scalar equations that result from the matrix equation A+S=0 for the 9 unknowns in SIG:

```
> solve(convert(A+S,set),convert(SIG,set)):
> assign(%);
```

Determine the first term of the asymptotic approximation of each of the elements of the matrix SIG as α1 becomes large.

```
> s11a:=op(1,asympt(s11,alpha1));
  s12a:=op(1,asympt(s12,alpha1));
  s13a:=op(1,asympt(s13,alpha1));
  s21a:=op(1,asympt(s21,alpha1));
  s22a:=simplify(op(1,asympt(s22,alpha1)));
  s23a:=op(1,asympt(s23,alpha1));
  s31a:=op(1,asympt(s31,alpha1));
  s32a:=op(1,asympt(s32,alpha1));
  s33a:=op(1,asympt(s33,alpha1));
```

$$s11a := \frac{\alpha 1}{\alpha 2\, R0^2}$$

$$s12a := -\frac{1}{R0}$$

$$s13a := -\frac{\alpha 1}{\alpha 2\, R0^2}$$

$$s21a := -\frac{1}{R0}$$

$$s22a := \frac{R0-1}{R0^2}$$

$$s23a := \frac{1}{R0^2}$$

$$s31a := -\frac{\alpha 1}{\alpha 2\, R0^2}$$

$$s32a := \frac{1}{R0^2}$$

$$s33a := \frac{\alpha 1}{\alpha 2\, R0^2} \quad \textbf{(15)}$$

The Expectation of S+I+R, divided by N, is denoted Expsum:

```
> Expsum:=simplify(x10+x20+x30);
```

$$Expsum := 1 \quad \textbf{(16)}$$

The Variance of S+I+R, divided by N, is denotede Varsum:

```
> Varsum:=simplify(s11+s22+s33+2*s12+2*s13+2*s23);
```

$$Varsum := 1 \quad \textbf{(17)}$$

I define ρI as the first term in the asymptotic approximation of the ratio x20*N/ $\sqrt{s22 \cdot N}$ for large α1.

```
> x20;
  x20a:=op(1,asympt(x20,alpha1));
  rhoI:=simplify(x20a*N/sqrt(s22a*N)) assuming R0>1;
```

$$\frac{\alpha 2\,(R0-1)}{R0\,(\alpha 2+\alpha 1-1)}$$

$$x20a := \frac{\alpha 2\,(R0-1)}{R0\,\alpha 1}$$

$$rhoI := \frac{\alpha 2\sqrt{R0-1}\,\sqrt{N}}{\alpha 1} \quad (18)$$

This expression for ρI is the same as for sirs2c and sirs3c.

```
> s11;
```

$$\frac{\alpha 2\,\alpha 1\,R0^2+R0-2\,\alpha 1\,R0+\alpha 1^2\,R0-2\,R0\,\alpha 2+2\,\alpha 2\,\alpha 1\,R0+R0\,\alpha 2^2+\alpha 2^2-\alpha 2}{\alpha 2\,R0^2\,(\alpha 2+\alpha 1\,R0-1)} \quad (19)$$

```
> alpha:=alpha1+alpha2-1;
```

$$\alpha := \alpha 2+\alpha 1-1 \quad (20)$$

```
> R1:=(alpha1*R0+alpha2-1)/alpha;
```

$$R1 := \frac{\alpha 2+\alpha 1\,R0-1}{\alpha 2+\alpha 1-1} \quad (21)$$

```
> s22;
```

$$\Big(\alpha 1^3\,R0^2-R0\,\alpha 1^3-\alpha 2\,\alpha 1^2\,R0+R0^3\,\alpha 1^2\,\alpha 2+\alpha 2\,\alpha 1^2\,R0^2+2\,\alpha 1^2\,R0-2\,\alpha 1^2\,R0^2 \quad (22)$$
$$+\alpha 2\,\alpha 1\,R0-\alpha 1\,R0-2\,\alpha 2\,\alpha 1\,R0^2+\alpha 1\,R0^2-R0^3\,\alpha 1\,\alpha 2+\alpha 2^2\,\alpha 1\,R0^3$$
$$+\alpha 2^2\,\alpha 1\,R0+\alpha 2^3+R0\,\alpha 2^2-2\,\alpha 2^2\,R0^2+\alpha 2\,R0^2-\alpha 2^2+\alpha 2^3\,R0^2-R0\,\alpha 2^3\Big)\Big/$$
$$\Big(\big(R0\,\alpha 1^3+2\,\alpha 2\,\alpha 1^2\,R0-2\,\alpha 1^2\,R0+\alpha 2^2\,\alpha 1\,R0-2\,\alpha 2\,\alpha 1\,R0+\alpha 1\,R0+\alpha 2\,\alpha 1^2$$
$$-\alpha 1^2+2\,\alpha 2^2\,\alpha 1-4\,\alpha 2\,\alpha 1+2\,\alpha 1-3\,\alpha 2^2+3\,\alpha 2-1+\alpha 2^3\big)\,R0^2\Big)$$

```
> s11num:=op(1,s11);
```

$$s11num := \alpha 2\,\alpha 1\,R0^2+R0-2\,\alpha 1\,R0+\alpha 1^2\,R0-2\,R0\,\alpha 2+2\,\alpha 2\,\alpha 1\,R0+R0\,\alpha 2^2 \quad (23)$$
$$+\alpha 2^2-\alpha 2$$

The numerator of s11 can be approximated by s11numa:

```
> s11numa:=(alpha1-1)^2*R0+alpha*alpha2*R1*(R0+2)-alpha2*(R0-1)-
  alpha2^2;
```

$$s11numa := (\alpha 1-1)^2\,R0+\alpha 2\,(\alpha 2+\alpha 1\,R0-1)\,(R0+2)-\alpha 2\,(R0-1)-\alpha 2^2 \quad (24)$$

```
> simplify(s11num-s11numa);
```

$$0 \quad (25)$$

An alternative expression for s11 is the following:

```
> s11alt:=(alpha1-1)^2/(alpha*alpha2*R0*R1) + (R0+2)/R0^2 - (R0+
```

```
alpha2-1)/(alpha*R0^2*R1);
```

$$s11alt := \frac{(\alpha 1-1)^2}{\alpha 2\, R0\,(\alpha 2+\alpha 1\, R0-1)} + \frac{R0+2}{R0^2} - \frac{R0+\alpha 2-1}{R0^2\,(\alpha 2+\alpha 1\, R0-1)} \quad (26)$$

```
> simplify(s11-s11alt);
```

$$0 \quad (27)$$

I give two-term asymptotic expansions of the elements of SIG for large alpha1:

```
> s11b:=map(simplify,convert(asympt(s11,alpha1,1),polynom));
s12b:=map(simplify,convert(asympt(s12,alpha1,2),polynom));
s13b:=map(simplify,convert(asympt(s13,alpha1,1),polynom));
s21b:=map(simplify,convert(asympt(s21,alpha1,2),polynom));
s22b:=map(simplify,convert(asympt(s22,alpha1,2),polynom));
s23b:=map(simplify,convert(asympt(s23,alpha1,2),polynom));
s31b:=map(simplify,convert(asympt(s31,alpha1,1),polynom));
s32b:=map(simplify,convert(asympt(s32,alpha1,2),polynom));
s33b:=map(simplify,convert(asympt(s33,alpha1,1),polynom));
```

$$s11b := \frac{\alpha 1}{\alpha 2\, R0^2} + \frac{\alpha 2\, R0^2 + 2\, R0\, \alpha 2 - 2\, R0 - \alpha 2 + 1}{\alpha 2\, R0^3}$$

$$s12b := -\frac{1}{R0} + \frac{\alpha 2\,(R0-1)}{\alpha 1\, R0^2}$$

$$s13b := -\frac{\alpha 1}{\alpha 2\, R0^2} + \frac{\alpha 2\, R0^2 - 2\, R0\, \alpha 2 + 2\, R0 + \alpha 2 - 1}{\alpha 2\, R0^3}$$

$$s21b := -\frac{1}{R0} + \frac{\alpha 2\,(R0-1)}{\alpha 1\, R0^2}$$

$$s22b := \frac{R0-1}{R0^2} + \frac{R0^3\, \alpha 2 - \alpha 2\, R0^2 + R0 - 1 + \alpha 2}{R0^3\, \alpha 1}$$

$$s23b := \frac{1}{R0^2} - \frac{\alpha 2\, R0^2 - R0\, \alpha 2 + R0 - 1 + \alpha 2}{R0^3\, \alpha 1}$$

$$s31b := -\frac{\alpha 1}{\alpha 2\, R0^2} + \frac{\alpha 2\, R0^2 - 2\, R0\, \alpha 2 + 2\, R0 + \alpha 2 - 1}{\alpha 2\, R0^3}$$

$$s32b := \frac{1}{R0^2} - \frac{\alpha 2\, R0^2 - R0\, \alpha 2 + R0 - 1 + \alpha 2}{R0^3\, \alpha 1}$$

$$s33b := \frac{\alpha 1}{\alpha 2\, R0^2} + \frac{-2\, \alpha 2\, R0^2 - 2\, R0 + R0\, \alpha 2 + R0^3\, \alpha 2 + 1 - \alpha 2}{R0^3\, \alpha 2} \quad (28)$$